

```
-----
;
;                               QSOxx.ASM
;
;                               QSOcorder
;
;                               Digital Audio "Instant Replay" Unit
;
;                               Dr. Steven E. Reyer, WA9VNJ
;
;                               Reyer & Associates
;                               P.O. Box 17821
;                               Milwaukee, WI 53217
;                               USA
;
;-----
;                               For personal, noncommercial use only.
;-----
;
; NOTE:  version number and date appear on
;        ROM page 3, and on the top of the
;        following lines.
;
; v. 2.3 - 10/17/93 - include NOPs for JNT1 on pwr-up
; v. 2.2 - 09/29/93 - remove unnecessary subroutines
; v. 2.1 - 08/10/93 - change manual reboot method
; v. 2.0 - 08/09/93 - add manual record mode
; v. 1.2 - 08/07/93 - include clear RAM routine
; v. 1.1 - 08/07/93 - shorten record loop to 42/55
; v. 1.0 - 08/06/93 - original program
;
;-----
;
; This program runs on an Intel 8748H (high speed version)
; microcontroller to allow "Instant Replay" of received
; communications quality audio.  It requires two 62256
; RAMs (32K x 8), an Analog Devices AD7569 8-bit audio
; I/O port, a 74LS374/574 demultiplexing latch, and
; analog antialiasing and reconstruction lowpass filters,
; and speaker driver amplifier, such as an LM380.
;
; The crystal required is 11 MHz (or 11.0592 MHz, which
; exceeds the specification of the 8748H, but not by very
; much, and should work fine in most instances).
;
; Three switches to control the system are connected as
; follows:
;
;     T0 (pin 1) - 10 second repeat pushbutton
;     T1 (pin 39)- 5 second repeat pushbutton
;     INT (pin 6) - Sample rate selector
;                   (open    = 13.41 kHz)
;                   (ground = 17.55 kHz)
;
; The sample rate determines the quality of the replayed
; audio, as well as the duration.  At the slow rate, the
; "10 second" switch plays for 9.8 seconds, and the
; "5 second" switch plays for 4.9 seconds.  At the fast
; rate, the times are 7.5 and 3.7 seconds.  At the slow
; rate, there will be some reconstruction noise, since
```

```

; the storage algorithm uses a /2 decimation, and a
; corresponding x2 interpolation, with simple 2-point
; averaging digital filtering. With the 3 kHz, 3rd order
; Chebyshev lowpass filters, this causes images from the
; sampling/decimation to be heard. A 3 kHz input signal
; will image to a resulting -30 dB at the fast rate, and
; -10 dB at the slow rate. Thus, the trade-off is between
; better quality vs. longer record time. Fortunately,
; ham radio applications will involve signals with little
; high frequency power.
;
; There are 42 states between samples at the fast rate,
; and 55 states at the slow rate. With the 11.0592 MHz
; clock, each state takes about 1.356 microseconds,
; = (15/11.0592) microseconds.
;
; The RAM address is multiplexed onto Port 1, and
; demultiplexed with the 74LS374 latch. The latch
; holds the MS bits, and Port 1 holds the LS bits.
; The MS bit of the latch operates the overload LED.
; Data transfers are made on the "BUS." Port 2 is used
; to construct the control signals as follows:
;
; Port bit      Pin      Signal
; P27           38      CS1- to RAM 1
; P26           37      CS0- to RAM 0
; P25           36      WE-  to RAMs
; P24           35      OE-  to RAMs
; P23           24      WR-  to AD7569
; P22           23      ST-  to AD7569
; P21           22      RD-  to AD7569
; P20           21      LATCH- to 74LS374
;
; Note: ST- starts the A/D converter on the falling edge.
;       It is a normally-low signal.
;
; Registers used are as follows:
;
; R7 - MS start address in buffer/temp constant 0
; R6 - LS start address in buffer
; R5 - MS end address
; R4 - Temporary sample storage
; R3 - Temporary sample storage
; R2 - MS current address pointer
; R1 - LS current address pointer
; R0 - sample average
;
; There are two modes of operation in the QSOcorder:
;
; AUTOMATIC MODE:
; -----
; When a "play" is requested, R7/R6 hold the computed
; starting "play address." This is either 5 seconds
; or 10 seconds earlier than the present buffer address.
; R5 is used to hold the MS end address, so the "play"
; knows when to stop and return to the "start" address
; to continue playing. R2/R1 are used as a temporary
; address during playing. When playing is no longer
; requested, R5/R6 are put into R2/R1 to continue
; recording. This will cause recording to begin at the

```

```

; same place it was halted when the play request was made.
; Playing continues as long as a play button is pressed.
;
; An "Overload LED" is connected to the latch MSB and is
; used to set the receiver audio gain. Adjust the
; receiver audio until the LED just flickers, and then
; back-off the setting a small amount. Too much receiver
; audio will cause some distortion, and too little will
; allow quantizing noise to be heard.
;
;
; MANUAL MODE:
; -----
; Manual mode is entered by holding the 10-second push-
; button down while powering-up the system. It will then
; enter the 1 second memory clear routine, and finally
; enter manual mode and light up the LED. This indicates
; it is idling, and ready for a record command. The record
; command consists of a quick press of the 10 second button.
; The LED will extinguish, and a 10 second recording will
; be made, after which the system lights up the LED and
; awaits a play command. This is similarly caused by a quick
; press of the 10 second button. It will play the complete
; 10 seconds, and return to the play-idle state, awaiting
; another play command. Whenever the LED is on, in either
; idle state, the mode may be cancelled by a quick press
; of the 5 second button. The system will then enter the
; automatic mode immediately. The manual mode can only
; be reentered by the power-up technique described above.
;
; The playback sample rate in manual mode will vary
; depending on whether the 10 second button is being
; pressed. This will also happen during record mode.
; The variation is from 17.55 kHz to 16.75 kHz which
; causes a noticeable change on music but is barely
; detectable on voice.
;
; There is a 1 second delay (with LED on) upon power-up
; in either mode, for the RAM-clear routine.
;
;
;-----
;----- Memory Page 0 -----
;-----

```

```

0000          ORG      0000H          ;Start of EPROM memory

0000 15      START   DIS      I          ;Turn off external interrupt
0001 35          DIS      TCNTI        ;...and timer counter int
0002 5400     CALL    INIT            ;Initialize ports and data
0004 26CB     JNT0    MANA            ;Manual record mode request
0006 00          NOP                    ;Room for JNT1 on power up
0007 00          NOP                    ;Same
0008 541C     CALL    CLEAR           ;Clear all of RAM memory

```

```

;----- Using RAM chip 0 -----
;----- RECORD -----
;Reset the start address to previous end address
000A 76B7    RECOX   JF1      PLAY0A    ;If manual, keep playing
000C FD      RECMAN  MOV      A,R5      ;Get MS end addr byte
000D AA          MOV      R2,A        ;Put into counter
000E FE          MOV      A,R6      ;Get LS end addr byte

```

```

000F A9          MOV      R1,A          ;Put into counter
0010 BF00        MOV      R7,#0        ;Constant zero

;Start the A/D converter
0012 8A04        RECO    ORL      P2,#00000100B    ;ST- = 1          2 states
0014 9AFB        ANL      P2,#11111011B    ;ST- = 0          2 A/D
0016 861A        JNI      FAST0A          ;Fast sampling?   2
0018 5498        CALL     WAIT13         ;No, add 13 states (13)

;Read the A/D converter into accum
001A 81          FAST0A  MOVX     A,@R1          ;Dummy to reset BUS  2
001B 9AFD        ANL      P2,#11111011B    ;RD- = 0          2
001D 08          INS      A,BUS          ;Read A/D on BUS   2
001E 8A02        ORL      P2,#00000010B    ;RD- = 1          2

;Send accum data to D/A converter
0020 02          OUTL     BUS,A          ;Send data to BUS   2
0021 9AB6        ANL      P2,#10110110B    ;WR-, LATCH-, CS0==0 2
0023 8A08        ORL      P2,#00001000B    ;WR- = 1          2 D/A
0025 AC          MOV      R4,A          ;Save sample        1

;Check whether overload is occurring. F0=1=yes.
0026 FA          MOV      A,R2          ;MS byte addr      1
0027 537F        ANL      A,#01111111B    ;Turn LED on        2
0029 B62F        JF0      ON0A          ;Should be on?     2
002B 4380        ORL      A,#10000000B    ;No, turn off       2
002D 0433        JMP      OFF0A         ;Continue           2
002F 00          ON0A    NOP           ;Pad timing         (1)
0030 00          NOP           ;                   (1)
0031 00          NOP           ;                   (1)
0032 00          NOP           ;                   (1)

;Put sample average into RAM at current address
;Send address to latches
0033 39          OFF0A   OUTL     P1,A          ;Out to port 1     2
0034 8A01        ORL      P2,#00000001B    ;LATCH- = 1        2
0036 F9          MOV      A,R1          ;LS byte addr      1
0037 39          OUTL     P1,A          ;Out to port 1     2

;Send data to RAM
0038 F8          MOV      A,R0          ;Get sample average 1
0039 02          OUTL     BUS,A          ;Put onto BUS      2
003A 9ADF        ANL      P2,#11011111B    ;WE- = 0          2
003C 8A24        ORL      P2,#00100100B    ;WE--=1, ST--=1    2

;Start the A/D converter
003E 9AFB        ANL      P2,#11111011B    ;ST- = 0          2 A/D
0040 8644        JNI      FAST0B          ;Fast sampling?     2
0042 5498        CALL     WAIT13         ;No, add 13 states (13)

;Read the A/D converter into accum
0044 81          FAST0B  MOVX     A,@R1          ;Dummy to reset BUS  2
0045 9AFD        ANL      P2,#11111011B    ;RD- = 0          2
0047 08          INS      A,BUS          ;Read A/D data on BUS 2
0048 8A42        ORL      P2,#01000010B    ;RD--=1;CS0--=1    2

;Send accum data to D/A converter
004A 02          OUTL     BUS,A          ;Send data to BUS   2
004B 9AF7        ANL      P2,#11110111B    ;WR- = 0          2
004D 8A08        ORL      P2,#00001000B    ;WR- = 1          2 D/A

```

```

;Check for overload - set F0 if yes.
004F 85          CLR      F0          ;LED flag off          1
0050 9655       JNZ      OFF0B        ;Overload?            2
0052 95         CPL      F0          ;Yes! Turn on        (1)
0053 0458       JMP      ON0B          ;Continue             (2)
0055 00         OFF0B  NOP          ;Pad timing          1
0056 00         NOP          ;                      1
0057 00         NOP          ;                      1

;Find the average
0058 6C         ON0B  ADD      A,R4        ;Add old sample      1
0059 67         RRC      A          ;Average the two     1
005A A8         MOV      R0,A        ;Save for later      1

;Increment address counters in R2:R1
005B F9         MOV      A,R1          ;Get LS byte         1
005C 0301       ADD      A,#1          ;Bump LS byte        2
005E A9         MOV      R1,A          ;Replace             1
005F FA         MOV      A,R2          ;Get MS byte         1
0060 7F         ADDC     A,R7          ;Add carry + (R7=0)  1
0061 AA         MOV      R2,A          ;Replace             1
0062 F2C7       JB7     TEST1A        ;Change RAM?        2

0064 566C       TEST0  JT1     TA          ;T1 up? Check T0    2
0066 FA         MOV      A,R2          ;No, fix start addr
0067 D380       XRL     A,#10000000B  ;Addr 5 secs back
0069 AF         MOV      R7,A          ;Save as MS start
006A 0472       JMP      PL0          ;Now play

006C 3612       TA     JT0     REC0        ;T0 up? Keep rec.   2
006E 7612       JF1     REC0        ;Manual, ignore T0

0070 FA         MOV      A,R2          ;Addr 10 secs back
0071 AF         MOV      R7,A          ;Save as MS start

0072 F9         PL0    MOV      A,R1          ;LS byte addr
0073 AE         MOV      R6,A          ;Save LS start/end
0074 FA         MOV      A,R2          ;Get MS addr end
0075 AD         MOV      R5,A          ;Save as MS end

;Set starting address for playing
0076 FF         START0 MOV      A,R7          ;Get MS start
0077 AA         MOV      R2,A          ;Into counter
0078 FE         MOV      A,R6          ;Get LS start
0079 A9         MOV      R1,A          ;Into counter

;----- PLAY -----
;Get a byte from current address into accum
;Send address to latches
007A 81         PLAY0  MOVX     A,@R1        ;Dummy to reset BUS  2
007B 9ABF       ANL     P2,#10111111B  ;CS0- = 0            2
007D FA         MOV      A,R2          ;MS byte addr        1
007E 4380       ORL     A,#10000000B  ;Turn off LED        2
0080 39         OUTL    P1,A          ;Out to port 1       2
0081 9AFE       ANL     P2,#11111110B  ;LATCH- = 0          2
0083 8A01       ORL     P2,#00000001B  ;LATCH- = 1          2
0085 F9         MOV      A,R1          ;LS byte addr        1
0086 39         OUTL    P1,A          ;Out to port 1       2
0087 9AEF       ANL     P2,#11101111B  ;OE- = 0             2

```

```

0089 08          INS      A,BUS          ;Read RAM data on BUS 2
008A 8A10        ORL      P2,#00010000B      ;OE- = 1 2
008C 8A40        ORL      P2,#01000000B      ;CS0- = 1 2

;Find the average
008E AB          MOV      R3,A          ;Save sample for now 1
008F 6C          ADD      A,R4          ;R4 = old sample 1
0090 67          RRC      A          ;Find average 1
0091 00          NOP                      ;Pad timing 1
0092 00          NOP                      ; 1
0093 00          NOP                      ; 1
0094 00          NOP                      ; 1
0095 00          NOP                      ; 1

;Send accum data to D/A converter
0096 869A        JNI      FAST0C        ;Fast sampling? 2
0098 5498        CALL     WAIT13        ;No, add 13 states (13)
009A 02          FAST0C  OUTL     BUS,A  ;Send data to BUS 2
009B 9AF7        ANL     P2,#11110111B ;WR- = 0 2
009D 8A08        ORL     P2,#00001000B ;WR- = 1 2 D/A

;Increment address counters in R2:R1
009F F9          MOV      A,R1          ;Get LS byte 1
00A0 0301        ADD     A,#1          ;Bump LS byte 2
00A2 A9          MOV     R1,A          ;Replace 1
00A3 FA          MOV     A,R2          ;Get MS byte 1
00A4 1300        ADDC   A,#0          ;Add carry 2
00A6 AA          MOV     R2,A          ;Replace 1
00A7 F2C9        JB7    TST1A         ;Change RAM? 2

00A9 F9          MOV     A,R1          ;Get LS addr 1
00AA DE          XRL    A,R6          ;At end? 1
00AB 96B1        JNZ    TST0          ;Not yet 2

00AD FA          MOV     A,R2          ;Get MS addr 1
00AE DD          XRL    A,R5          ;At end? 1
00AF C676        JZ     START0        ;Yes, restart

00B1 56B5        TST0   JT1    TB          ;T1 up? 2
00B3 04B7        JMP    PLAY0A        ;No, play sound 2

00B5 360A        TB     JT0    REC0X      ;Finished playing?

00B7 FB          PLAY0A MOV     A,R3          ;Recall new sample 1
00B8 AC          MOV     R4,A          ;Make into old 1
00B9 5498        CALL   WAIT13        ;Wait 13 states 13
00BB 00          NOP                      ;Pad timing 1

;Send sample to D/A converter
00BC 86C0        JNI     FAST0D        ;Fast sampling? 2
00BE 5498        CALL   WAIT13        ;No, add 13 states (13)
00C0 02          FAST0D OUTL     BUS,A  ;Send data to BUS 2
00C1 9AF7        ANL    P2,#11110111B ;WR- = 0 2
00C3 8A08        ORL    P2,#00001000B ;WR- = 1 2 D/A
00C5 047A        JMP    PLAY0          ;Keep playing 2

00C7 245E        TEST1A JMP     TEST1        ;Cross page jumps
00C9 24AD        TST1A  JMP     TST1
00CB 4440        MANA   JMP     MANUAL

```

```

;-----
;----- Memory Page 1 -----
;-----
0100          ORG          0100H

;----- Using RAM chip 1 -----
;----- RECORD -----
;Reset the start address to previous end address
0100 76B3  REC1X  JF1      PLAY1A          ;If manual, keep playing
0102 FD          MOV      A,R5          ;Get MS end addr byte
0103 AA          MOV      R2,A          ;Put into counter
0104 FE          MOV      A,R6          ;Get LS end addr byte
0105 A9          MOV      R1,A          ;Put into counter
0106 BF00        MOV      R7,#0         ;Constant zero

;Start the A/D converter
0108 8A04  REC1   ORL      P2,#00000100B ;ST- = 1          2
010A 9AFB          ANL      P2,#11111011B ;ST- = 0          2 A/D
010C 8610          JNI      FAST1A         ;Fast sampling?  2
010E 5498          CALL     WAIT13        ;No, add 13 states (13)

;Read the A/D converter into accum
0110 81          FAST1A MOVX     A,@R1          ;Dummy to reset BUS  2
0111 9AFD          ANL      P2,#11111101B        ;RD- = 0          2
0113 08          INS      A,BUS          ;Read A/D data on BUS 2
0114 8A02          ORL      P2,#00000010B        ;RD- = 1          2

;Send accum data to D/A converter
0116 02          OUTL     BUS,A          ;Send data to BUS    2
0117 9A76          ANL      P2,#01110110B        ;WR-, LATCH-, CS1==0 2
0119 8A08          ORL      P2,#00001000B        ;WR- = 1          2 D/A
011B AC          MOV      R4,A          ;Save sample        1

;Check whether overload is occurring. F0=1=yes.
011C FA          MOV      A,R2          ;MS byte addr      1
011D 537F          ANL      A,#01111111B        ;Turn LED on        2
011F B625          JF0      ON1A          ;Should be on?     2
0121 4380          ORL      A,#10000000B        ;No, turn off       2
0123 2429          JMP      OFF1A         ;Continue           2
0125 00          ON1A   NOP            ;                   (1)
0126 00          NOP            ;                   (1)
0127 00          NOP            ;                   (1)
0128 00          NOP            ;                   (1)

;Put sample average into RAM at current address
;Send address to latches
0129 39          OFF1A  OUTL     P1,A          ;Out to port 1     2
012A 8A01          ORL      P2,#00000001B        ;LATCH- = 1        2
012C F9          MOV      A,R1          ;LS byte addr      1
012D 39          OUTL     P1,A          ;Out to port 1     2

;Send data to RAM
012E F8          MOV      A,R0          ;Get sample average 1
012F 02          OUTL     BUS,A          ;Put onto BUS      2
0130 9ADF          ANL      P2,#11011111B        ;WE- = 0          2
0132 8A24          ORL      P2,#00100100B        ;WE-=1, ST-=1     2

;Start the A/D converter
0134 9AFB          ANL      P2,#11111011B        ;ST- = 0          2 A/D
0136 863A          JNI      FAST1B         ;Fast sampling?    2

```

```

0138 5498          CALL      WAIT13          ;No, add 13 states (13)

;Read the A/D converter into accum
013A 81          FAST1B  MOVX      A,@R1          ;Dummy to reset BUS 2
013B 9AFD          ANL      P2,#11111101B      ;RD- = 0 2
013D 08          INS      A,BUS          ;Read A/D data on BUS 2
013E 8A82          ORL      P2,#10000010B      ;RD-=1;CS1-=1 2

;Send accum data to D/A converter
0140 02          OUTL     BUS,A          ;Send data to BUS 2
0141 9AF7          ANL      P2,#11110111B      ;WR- = 0 2
0143 8A08          ORL      P2,#00001000B      ;WR- = 1 2 D/A

;Check for overload - set F0 if yes.
0145 85          CLR      F0          ;LED flag off 1
0146 964B          JNZ      OFF1B        ;Overload? 2
0148 95          CPL      F0          ;Yes! Turn on (1)
0149 244E          JMP      ON1B        ;Continue (2)
014B 00          OFF1B  NOP          ;Pad timing 1
014C 00          NOP          ; 1
014D 00          NOP          ; 1

;Find the average
014E 6C          ON1B   ADD      A,R4          ;Add old value 1
014F 67          RRC      A          ;Average the two 1
0150 A8          MOV      R0,A          ;Save for later 1

;Increment address counters in R2:R1
0151 F9          MOV      A,R1          ;Get LS byte 1
0152 0301          ADD      A,#1          ;Bump LS byte 2
0154 A9          MOV      R1,A          ;Replace 1
0155 FA          MOV      A,R2          ;Get MS byte 1
0156 7F          ADDC     A,R7          ;Add carry + (R7=0) 1
0157 AA          MOV      R2,A          ;Replace 1
0158 F25E          JB7     TEST1        ;Change RAM? 2
015A 76C3          JF1     MAN2A        ;Manual, stop rec.
015C 0464          JMP      TEST0        ;Yes

015E 5666          TEST1   JT1     TC          ;T1 up? Check T0 2
0160 FA          MOV      A,R2          ;No, fix start addr
0161 D380          XRL     A,#10000000B  ;Addr 5 secs back
0163 AF          MOV      R7,A          ;Save as MS start
0164 246A          JMP      PL1          ;Now play

0166 3608          TC     JT0     REC1        ;T0 up? Keep rec. 2

0168 FA          MOV      A,R2          ;Addr 10 secs back
0169 AF          MOV      R7,A          ;Save as MS start

016A F9          PL1    MOV      A,R1          ;LS byte addr
016B AE          MOV      R6,A          ;Save LS start/end
016C FA          MOV      A,R2          ;Get MS addr end
016D AD          MOV      R5,A          ;Save as MS end

;Set starting address for playing
016E FF          START1  MOV      A,R7          ;Get MS start
016F AA          MOV      R2,A          ;Into counter
0170 FE          MOV      A,R6          ;Get LS start
0171 A9          MOV      R1,A          ;Into counter

```



```

;----- PLAY -----
;Get a byte from current address into accum
;Send address to latches
0172 81      PLAY1  MOVX    A,@R1      ;Dummy to reset BUS  2
0173 9A7F    ANL     P2,#01111111B ;CS1- = 0           2
0175 FA      MOV     A,R2      ;MS byte addr       1
0176 4380    ORL     A,#10000000B ;Turn off LED       2
0178 39      OUTL    P1,A      ;Out to port 1     2
0179 9AFE    ANL     P2,#11111110B ;LATCH- = 0        2
017B 8A01    ORL     P2,#00000001B ;LATCH- = 1        2
017D F9      MOV     A,R1      ;LS byte addr       1
017E 39      OUTL    P1,A      ;Out to port 1     2
017F 9AEF    ANL     P2,#11101111B ;OE- = 0           2
0181 08      INS     A,BUS     ;Read RAM data on BUS 2
0182 8A10    ORL     P2,#00010000B ;OE- = 1           2
0184 8A80    ORL     P2,#10000000B ;CS1- = 1           2

;Find the average
0186 AB      MOV     R3,A      ;Save sample for now 1
0187 6C      ADD     A,R4      ;R4 = old sample    1
0188 67      RRC     A        ;Find average       1
0189 00      NOP                    ;Pad timing         1
018A 00      NOP                    ;                   1
018B 00      NOP                    ;                   1
018C 00      NOP                    ;                   1
018D 00      NOP                    ;                   1

;Send accum data to D/A converter
018E 8692    JNI     FAST1C    ;Fast sampling?     2
0190 5498    CALL    WAIT13   ;No, add 13 states (13)
0192 02      FAST1C OUTL    BUS,A ;Send data to BUS   2
0193 9AF7    ANL     P2,#11110111B ;WR- = 0           2
0195 8A08    ORL     P2,#00001000B ;WR- = 1           2 D/A

;Increment address counters in R2:R1
0197 F9      MOV     A,R1      ;Get LS byte       1
0198 0301    ADD     A,#1      ;Bump LS byte      2
019A A9      MOV     R1,A      ;Replace           1
019B FA      MOV     A,R2      ;Get MS byte       1
019C 1300    ADDC   A,#0      ;Add carry         2
019E AA      MOV     R2,A      ;Replace           1
019F F2A5    JB7     TE        ;Change RAM?       2
01A1 76C3    JF1     MAN2A     ;Manual, stop play
01A3 04B1    JMP     TST0      ;Yes

01A5 F9      TE      MOV     A,R1      ;Get LS addr       1
01A6 DE      XRL    A,R6      ;At end?           1
01A7 96AD    JNZ    TST1      ;Not yet           2

01A9 FA      MOV     A,R2      ;Get MS addr       1
01AA DD      XRL    A,R5      ;At end?           1
01AB C66E    JZ     START1    ;Yes, restart      2

01AD 56B1    TST1   JT1     TD      ;T1 up?           2
01AF 24B3    JMP     PLAY1A   ;No, play sound    2

01B1 3600    TD      JT0     REC1X   ;Finished playing?

01B3 FB      PLAY1A MOV    A,R3      ;Recall new sample 1
01B4 AC      MOV    R4,A      ;Make into old     1

```

```

01B5 5498          CALL    WAIT13          ;Wait 13 states      13
01B7 00           NOP                      ;Pad timing         1

;Send sample to D/A converter
01B8 86BC          JN1     FAST1D          ;Fast sampling?     2
01BA 5498          CALL    WAIT13          ;No, add 13 states  (13)
01BC 02           FAST1D  OUTL    BUS,A    ;Send data to BUS   2
01BD 9AF7          ANL    P2,#11110111B   ;WR- = 0            2
01BF 8A08          ORL    P2,#00001000B   ;WR- = 1            2 D/A
01C1 2472          JMP    PLAY1           ;Keep playing       2

01C3 4454          MAN2A  JMP    MAN2           ;Cross page jumps

;-----
;----- Memory Page 2 -----
;-----

0200              ORG    0200H

;-----
0200  INIT          ;Initialize ports, BUS, and memories
0200 C5           SEL    RB0             ;Use RAM register bank 0
0201 E5           SEL    MB0             ;...and EPROM mem bank 0
0202 23FF          MOV    A,#11111111B    ;All ones
0204 39           OUTL   P1,A           ;Port 1 = 11111111B
0205 02           OUTL   BUS,A          ;BUS = 11111111B
0206 23FB          MOV    A,#11111011B    ;ST- idles low
0208 3A           OUTL   P2,A           ;Port 2 = 11111011B
0209 BF00          MOV    R7,#0           ;MS start addr
020B BE00          MOV    R6,#0           ;LS start addr
020D BD00          MOV    R5,#0           ;MS end addr
020F BC00          MOV    R4,#0           ;Temp sample storage
0211 BB00          MOV    R3,#0           ;Temp sample storage
0213 BA00          MOV    R2,#0           ;MS curr addr
0215 B900          MOV    R1,#0           ;LS curr addr
0217 B87F          MOV    R0,#127         ;sample average
0219 85           CLR    F0              ;Overload LED off
021A A5           CLR    F1              ;Automatic mode
021B 93           RETR

;-----
021C  CLEAR          ;Clear RAM memory, two bytes at a time
021C 237F          MOV    A,#127          ;"zero" sample voltage
021E 02           OUTL   BUS,A          ;Send to BUS/RAM
021F BA00          MOV    R2,#00000000B   ;MS address = 0; LED on
0221 B900          MOV    R1,#00000000B   ;LS address = 0
0223 9A3F          ANL    P2,#00111111B   ;CS0-,CS1-=0

0225 FA           MORCLR MOV    A,R2           ;MS address
0226 39           OUTL   P1,A           ;To latch
0227 9AFE          ANL    P2,#11111110B   ;LATCH-=0
0229 8A01          ORL    P2,#00000001B   ;LATCH-=1
022B F9           MOV    A,R1           ;LS address
022C 39           OUTL   P1,A           ;To port 1
022D 9ADF          ANL    P2,#11011111B   ;WE-=0
022F 8A20          ORL    P2,#00100000B   ;WE-=1

0231 F9           MOV    A,R1           ;LS address
0232 0301          ADD    A,#1           ;Bump address
0234 A9           MOV    R1,A           ;Replace
0235 FA           MOV    A,R2           ;MS address

```

```

0236 1300          ADDC    A,#0           ;Add carry
0238 AA           MOV     R2,A           ;Replace
0239 F23D         JB7     NOCLR          ;More than 32k ?
023B 4425         JMP     MORCLR          ;Continue

023D 8AC0         NOCLR   ORL     P2,#11000000B      ;CS0-,CS1-=1
023F 93          RETR     RETR          ;Finished

;-----
0240             MANUAL   ;Manual record/play mode requested
0240 541C         CALL    CLEAR          ;Clear all of RAM memory
0242 2642         MAN3    JNT0    MAN3          ;Wait for key up
0244 5484         CALL    DEBOUN         ;Debounce key
0246 5474         CALL    LEDON          ;Turn on LED

;Get a sample and output to DAC
0248 5464         WAIT1   CALL    SAMPLE          ;A/D and D/A
024A 465E         JNT1    CANCEL          ;Press T1 to cancel
024C 3648         JT0     WAIT1          ;T0 up, wait

024E 547C         CALL    LEDOFF         ;Turn off LED
0250 A5           CLR     F1             ;Ready to set F1
0251 B5           CPL     F1             ;F1=1, LED is off
0252 040C         JMP     RECMAN         ;Record from start

0254             MAN2    ;Back from recording or playing 64k
0254 5474         CALL    LEDON          ;Turn on LED

;Get a sample and output to DAC
0256 5464         WAIT2   CALL    SAMPLE          ;A/D and D/A
0258 465E         JNT1    CANCEL          ;Press T1 to cancel
025A 3656         JT0     WAIT2          ;Wait for T0 press

025C 047A         JMP     PLAY0          ;Now play 64k

025E             CANCEL   ;T1 was pressed while in Memory Idle modes.
025E             ;Cancel Memory mode.
025E 5400         CALL    INIT           ;Initialize & clear flag
0260 547C         CALL    LEDOFF         ;Turn off LED
0262 040A         JMP     RECOX          ;"Warm" reboot

;-----
0264             SAMPLE   ;Do A/D and D/A conversion
0264 8A04         ORL     P2,#00000100B      ;ST- = 1
0266 9AFB         ANL     P2,#11111011B      ;ST- = 0
0268 81          MOVX    A,@R1           ;Dummy to reset BUS
0269 9AFD         ANL     P2,#11111101B      ;RD- = 0
026B 08          INS     A,BUS           ;Read A/D on BUS
026C 8A02         ORL     P2,#00000010B      ;RD- = 1
026E 02          OUTL   BUS,A           ;Send data to BUS
026F 9AF7         ANL     P2,#11110111B      ;WR- = 0
0271 8A08         ORL     P2,#00001000B      ;WR- = 1
0273 93          RETR     RETR          ;Finished

;-----
0274             LEDON    ;Turn on LED
0274 537F         ANL     A,#01111111B      ;Turn on LED
0276 39          OUTL   P1,A           ;Out to port 1
0277 9AFE         ANL     P2,#11111110B      ;LATCH- = 0
0279 8A01         ORL     P2,#00000001B      ;LATCH- = 1

```

027B 93

RETR

```

;-----
027C      LEDOFF      ;Turn off LED
027C 4380      ORL      A,#10000000B      ;Turn off LED
027E 39        OUTL     P1,A              ;Out to port 1
027F 9AFE      ANL      P2,#11111110B    ;LATCH- = 0
0281 8A01      ORL      P2,#00000001B    ;LATCH- = 1
0283 93        RETR

```

```

;-----
0284      DEBOUN     ;Wait 50 milliseconds
0284 548F      CALL     MS10              ;Wait 10 ms.
0286 548F      CALL     MS10              ;Wait 10 ms.
0288 548F      CALL     MS10              ;Wait 10 ms.
028A 548F      CALL     MS10              ;Wait 10 ms.
028C 548F      CALL     MS10              ;Wait 10 ms.
028E 93        RETR

```

```

;-----
028F      MS10      ;Use timer to time 10 ms., approximately.
028F      ;Timer counts at a rate of xtal/15/32 Hz
028F      ;= 23.04 kHz (43.4028 us/count).
028F 231A      MOV      A,#26            ;230 counts
0291 62        MOV      T,A             ;Into timer
0292 55        STRT     T                ;Start timer
0293 1697      NOTFIN   JTF      TIMFIN   ;Finished
0295 4493      JMP      NOTFIN
0297 93        TIMFIN  RETR

```

```

;-----
0298      WAIT13    ;Wait 13 states including CALL and RETR
0298 00        NOP
0299 00        NOP
029A 00        NOP
029B 00        NOP
029C 00        NOP
029D 00        NOP
029E 00        NOP
029F 00        NOP
02A0 00        NOP
02A1 93        RETR

```

```

;-----
;----- Memory Page 3 -----
;-----

```

```

;-----
03B0      ORG      03B0H
03B0      NOTICE   ;Copyright notice takes from 03B0H to 03FEH
03B0 20286329   DB      '(c) copyright 1993,'
03C4 20526579   DB      ' Reyer & Associates,'
03D8 20574139   DB      ' WA9VNJ QSOcorder'
03E9 20766572   DB      ' version 2.3'
03F5 2031302F   DB      ' 10/17/93'
03FE 20         DB      ' '
0000      END

```

